


中学计算机辅助读物

BASIC

编程 辅导



江苏教育出版社

TP312BA/
130

中学计算机辅助读物

BASIC 编程辅导

蔡绍稷 薛维明 编



江苏教育出版社

内 容 提 要

本书分两部分,第一部分,通过三十个例题,从输入、加工和输出三个方面,介绍了使用 BASIC 语言编写程序的基本方法;第二部分,以中学教材内容为基础,选编了八十例,各例中一般均有编程思路,同时还给出了如何用程序实现的方法。

本书可作为中学生和自学青年学习 BASIC 语言,提高编程能力的辅助读物,也可供中学教师进行计算机普及教学的参考资料。

中学计算机辅助读物

BASIC 编程辅导

蔡绍稷 薛维明 编

江苏教育出版社出版

江苏省新华书店发行 淮海印刷厂印刷
开本 787×1092毫米 1/32 印张 7.875 字数 166,000
1986年9月第1版 1986年9月第1次印刷

印数 1—3,750 册

书号: 7351·355 定价: 1.00 元

责任编辑 王建军

目 录

第一部分 编制BASIC程序的基本方法

一、数据的输入方式	2
二、数据的加工	9
1. BASIC 语言中允许使用的基本运算	9
2. 数据加工的过程	10
3. 数据加工中常见的几种基本类型	16
三、输出方式	45
1. 常规打印	45
2. 特定位置的打印	49
3. 高分辨率作图	72

第二部分 程序例选

一、质数	81
二、最大公因数和最小公倍数	86
三、集合运算、数制转换	94
四、方程	109
五、不等式	123
六、多项式	131

七、数列·····	138
八、排列与组合·····	146
九、函数计算·····	154
十、复数的运算·····	161
十一、定积分的近似计算·····	166
十二、统计·····	171
十三、距离、面积、体积·····	189
十四、打印图表·····	195
附录 磁盘文件的使用方法 ·····	214

第一部分 编制BASIC程序的基本方法

电子计算机,实质上是一个输入信息、加工信息、输出信息的机器。我们要让计算机完成某项任务,必须事先用一种语言,把该项任务按照一定的规则,编制一组有序的又能为计算机所接受的命令序列,再将这组命令序列输入到计算机中,机器执行这组命令序列,同时也就自动完成该项任务。BASIC语言是用来编写这些命令序列的工具之一。因此,编制一个BASIC程序,实际上就是把人们的意图(命令序列),通过BASIC语言表达出来(如果用FORTRAN语言表达出来,就是FORTRAN程序)。这些命令序列的内容一般可概括为如下三个方面:

1. 用什么方式、给哪些变量、输入哪些数据;
2. 对输入的数据怎样加工;
3. 用什么格式、输出哪些结果。

对于第一和第三两个方面,只要理解题意,根据不同的要求,选择恰当的输入和输出方式即可。困难往往来自第二个方面,其原因是:(1)对计算机的性能和特点了解不够;(2)对该问题的算法不清楚(如找不出运算规律)。下面就这三个方面来讨论编制BASIC程序的基本方法。

一、数据的输入方式

在编制程序时，首先遇到的问题是：如何将数据送给对应的变量。在 BASIC 语言中，给变量送值，可采用三种方式：计算赋值(LET)；键盘输入(INPUT)；成批赋值(READ/DATA)。

LET 语句的含义是：把等号右边表达式的值送到等号左边的变量中去。例如，

```
10 LET X = 123.4
```

```
20 LET Y = X * X + 6
```

```
30 LET N$ = "NANJING"
```

输入并执行上述语句后，计算机就会自动给数值变量(X、Y)和字符串变量(N\$)分配相应的存贮单元。这里，10 号语句的作用应理解为：把 123.4 这个数值，存放到变量 X 所对应的存贮单元中去。20 号语句的作用应理解为：首先计算表达式 $X^2 + 6$ 的值，然后，把计算得到的数值存放到变量 Y 所对应的存贮单元中。

计算机中的存贮器是用来存放程序 and 数据的，它具有破坏性读入的特性，如果没有新内容存入，该存贮单元的内容将一直保持不变。若有新内容存入，那么新内容将取代原内容。在编制程序时，常常要用到这一特性。例如，已知变量 X 和 Y 中分别存放了数值 10 和 20，现在要求交换它们的值，即变量 X 中的数值变成 20，而变量 Y 中的数值变为 10，初学者往往用如下语句完成：

```
10 LET X = Y
```

20 LET Y = X

我们先分析一下这两个语句的执行过程：10 号语句将变量 Y 的值(20)送到变量 X 所对应的存贮单元中，这样，根据破坏性读入的特性，变量 X 中的原值(10)就变成了新值(20)。所以，再执行 20 号语句时，将变量 X 的新值(20)又赋给了变量 Y，从而使变量 X 和变量 Y 中的值均为 20。因此，用这两个语句是不能达到交换目的的。从以上分析可知，没有达到交换目的的原因是，变量 X 的原值被破坏了。为了达到交换目的，就必须保存变量 X 中的原值。其方法是在 10 号语句给变量 X 赋新值前，先把 X 的原值存放在某一个备用变量中。用下列三个语句可以完成交换任务。

10 LET S = X

20 LET X = Y

30 LET Y = S

交换两个变量的内容，在编制程序中经常要用到，必须掌握这一基本方法。其书写特点是：头尾一样(备用变量名——工作单元)，中间两两一样(两个要交换内容的变量名)。

INPUT 语句，是在程序运行过程中(即敲入 RUN 命令后)，由人机对话方式输入数据的语句。一般用于修改程序中的参数，使所编程序具有灵活性和通用性。

READ/DATA 语句，一般用于成批数据的输入。用这种语句输入时，必须注意数据和变量的对应关系，即 DATA 语句中数据的排列顺序一定要和 READ 语句中的读入顺序一致。

例 1 已知三个数 A、B、C，要求按从大到小的顺序打印出来。

将 A、B、C 三个数,按从大到小的顺序排列,共有如下六种情况:

当 $A > B$ 且	$\left\{ \begin{array}{l} B > C \\ B \leq C \text{ 且} \end{array} \right.$	打印 ABC
		$\left\{ \begin{array}{l} A > C \text{ 打印 ACB} \\ A \leq C \text{ 打印 CAB} \end{array} \right.$
当 $A \leq B$ 且	$\left\{ \begin{array}{l} B < C \\ B \geq C \text{ 且} \end{array} \right.$	打印 CBA
		$\left\{ \begin{array}{l} A < C \text{ 打印 BCA} \\ A \geq C \text{ 打印 BAC} \end{array} \right.$

【程序及运行结果】

```

10 INPUT "A,B,C=";A,B,C
20 IF A>B THEN 110
30 IF B<C THEN 90
40 IF A<C THEN 70
50 PRINT B,A,C
60 GOTO 180
70 PRINT B,C,A
80 GOTO 180
90 PRINT C,B,A
100 GOTO 180
110 IF B>C THEN 170
120 IF A>C THEN 150
130 PRINT C,A,B
140 GOTO 180
150 PRINT A,C,B
160 GOTO 180

```

```
170 PRINT A,B,C
```

```
180 END
```

```
RUN
```

```
A,B,C=1,2,3
```

```
3
```

```
2
```

```
1
```

随着比较数据个数的增加，用上述方法编制的程序将会变得冗长而复杂。若改用下面的方法进行排序，可使程序变得简单。

(1) 不论 A、B、C 三个数的大小如何，先找出其中最大的数，将该数放入 A 所对应的存储单元中。具体步骤为：先比较 A 和 B，如果 $A < B$ ，则交换它们的内容，以保证 A 中为大数。然后再比较 A 和 C，如果 $A < C$ ，则将 A、C 的内容交换，从而使 A 中为三个数中的最大数。

(2) 比较 B 和 C，如果 $B > C$ ，则打印 ABC，否则打印 ACB。

【程序及运行结果】

```
10 INPUT "A,B,C=";A,B,C
```

```
20 IF A>B THEN 40
```

```
30 S=A:A=B:B=S
```

```
40 IF A>C THEN 60
```

```
50 S=A:A=C:C=S
```

```
60 IF B>C THEN 90
```

```
70 PRINT A,C,B
```

```
80 GOTO 100
```

```
90 PRINT A,B,C
```

100 END

RUN

A,B,C=1,2,3

3

2

1

从该程序中可看到,如果增加比较数据的个数,只要重复执行上述过程。

例 2 某班有 50 名学生,上学期学习三门课程。现要求编制一个查询程序,查询时,只要输入该学生的学号,屏幕上就能显示该学生的各课程成绩。

如果输入的数据较多,可使用 READ/DATA 语句输入。查询时,学号的输入方式应该灵活方便,为此,选用键盘输入语句来完成。

假设 50 位学生的学号分别为 1,2,3,...,50,并用变量 I 表示;三门课的成绩分别用变量 X、Y、Z 表示,要求查询学生的学号用变量 N 表示。为了完成查询任务,可设计如下一组命令序列:

(1) 按学号从小到大的顺序,将学生的学号和对应的三门课程的成绩数据,输入到计算机中的数据区;

(2) 输入待查学生的学号;

(3) 从数据区中,按学号从小到大的顺序每次取出一个学生的四个数据(I,X,Y,Z);

(4) 将取出的学号 I 和待查的学号 N 进行比较,如果相符,则表示已查到,并显示该学生的三门课程的成绩。否则继续取下一个学生的四个数据,再进行比较,重复该过程,直到相符为止。

下面以 5 位学生为例:

【程序及运行结果】

```
10 DATA 1,60,70,80,2,70,90,80
20 DATA 3,87,76,69,4,88,77,99
30 DATA 5,87,76,69,0,0,0,0
100 INPUT "N=";N
110 IF N=0 THEN 200
120 IF N>5 THEN 100
130 READ I,X,Y,Z
140 IF I=N THEN 160
150 GOTO 130
160 PRINT "I=";I;"X=";X;"Y=";Y;"Z=";Z
180 RESTORE
190 GOTO 100
200 END
```

RUN

N=2

I=2 X=70 Y=90 Z=80

N=5

I=5 X=87 Y=76 Z=69

N=0

【说明】

- (1) 10、20、30 号语句,将数据输入数据区;
- (2) 100 号语句,输入待查学生的学号;
- (3) 110 号语句,如果输入待查学号为 0,则结束查询。如

果超出查询范围,则由 120 号语句转 100 号语句重新查询;

(4) 130 号语句,从数据区中,取出某一位学生的一组数据 I, X, Y, Z;

(5) 140 号语句,表示如果待查学号和取出学号相符,则由 160 号语句,显示该学生的学号和成绩;

(6) 150 号语句,表示如果待查学号与取出学号不符,则转 130 号语句继续取下一组数据;

(7) 190 号语句,提供继续查询的功能。由于每次查询都要从 1 号学生开始比较,所以在继续查询前,由 180 号语句恢复数据区。

如果我们用二维数组 S 组织数据,第一个下标表示学号,第二个下标表示课程类别,如 S(1,1) 表示 1 号学生的第一门课程成绩,又如 S(2,3) 表示 2 号学生的第三门课程成绩,那么,编制的程序和运行结果如下。

【程序及运行结果】

```
10 DIM S(5,3)
20 FOR I=1 TO 5
30 FOR J=1 TO 3
40 READ S(I,J)
50 NEXT J
60 NEXT I
70 INPUT N
80 PRINT
90 IF N=0 THEN 140
100 FOR J=1 TO 3
110 PRINT S(N,J),
```

```

120 NEXT J
125 PRINT
130 GOTO 70
140 END

150 DATA 60,70,80,70,90,80
160 DATA 87,76,69,88,77,99
170 DATA 79,87,65

RUN
?2
70          90          80
?4
88          77          99
?0

```

二、数据的加工

计算机应用之广、影响之大是举世公认的,但是就计算机本身而言,它所提供的运算功能却是十分有限的。而在处理实际问题中,我们往往需要进行各种各样的运算,数据的加工就是把实际问题中,对数据进行的各种运算分解为一系列计算机能做的基本运算,然后用相应的语句组合这些运算,从而完成数据加工部分程序的编制。它是各种程序的主体部分。

1. BASIC语言中允许使用的基本运算

在计算机中,组成运算器的核心部件是加法器,因此,加

法操作是计算机最基本的运算功能，减法和乘除法是通过加法和移位实现的。此外，计算机必须具有逻辑判断功能，否则不能自动工作。于是，四则运算和逻辑运算就组成了计算机的基本运算。

在 BASIC 语言中，除了四则运算和逻辑运算外，为了给使用者提供方便，一般都配备了某些常用函数，它们是利用计算机的基本运算而编制成的一个个独立的子程序，只要写出它们的函数名及给出自变量值就可随时调用（基本 BASIC 语言中配备了十一个常用函数）。如乘方运算 $A \wedge B$ （或 $A \uparrow B$ ）是利用常用函数 e^x 和 $\ln x$ 来计算的，即先求 A 的对数 $\log(A)$ ，再乘 B 后求出其指数函数 $\exp(B * \log(A))$ 来实现的。括号运算只提供圆括号。以上就是编制基本 BASIC 程序中允许使用的运算及其函数。如果问题中需要用到其它运算或函数，则必须先进行变换。例如，计算 $5!$ ，必须变换为 $1 * 2 * 3 * 4 * 5$ ；计算 ctgx ，必须变换为 $1/\tan(x)$ 或 $\sin(x)/\cos(x)$ ；计算 $\int_a^b f(x)dx$ ，可选用矩形公式、梯形公式或抛物线等公式，将定积分的计算转化为一系列四则运算。

2. 数据加工的过程

完成数据加工的任务，一般需要以下四个步骤：

- (1) 了解问题的性质和任务；
- (2) 构成模型，选择算法；
- (3) 绘制描述加工过程的流程图；
- (4) 根据流程图，用相应的语句完成各框的功能。

其中(1)、(2)为编程准备工作，它取决于对实际问题的充

分了解和分析。(3)、(4)是具体编制程序的过程。下面我们通过一些例子来说明数据加工的过程。

例3 某班期终考三门课程,其中两门为主课,现要求挑选出符合下列条件之一的学生:

- 1° 三门课总分高于 270 分者;
- 2° 两门主课均在 95 分以上者;
- 3° 一门主课为 100 分,其它两门在 80 分以上者。

【分析】 本题是一个比较问题,要求把学生的成绩和挑选标准相比,满足条件者则被选中。我们设 x_1, x_2 为两门主课成绩, x_3 为副课成绩,学生的学号用 I 表示。

根据题意,挑选条件可用下面四个算式来表示:

$$a. x_1 + x_2 + x_3 > 270 \quad (\text{条件 1})$$

$$b. x_1 > 95, x_2 > 95 \quad (\text{条件 2})$$

$$c. x_1 = 100, x_2 > 80, x_3 > 80 \quad (\text{条件 3})$$

$$d. x_2 = 100, x_1 > 80, x_3 > 80 \quad (\text{条件 3})$$

可见只要满足其中一个算式者,即被选中。

【绘制流程图】 流程图是用各种几何图形、指向线及文字说明,来描述计算过程的有向图,它既能表达程序的设计思想,又便于检查、修改和阅读程序,它是编制程序的基本功。

【设计思路】 (1) 输入某学生的学号和三门课的成绩;

(2) 判断条件 1 是否满足? 如果满足则打印;

(3) 当条件 1 不满足时,再判条件 2 是否满足? 由于条件 2 要求 $x_1 > 95$ 和 $x_2 > 95$ 同时成立,所以,先判 $x_1 > 95$ 是否成立? 如果成立,再判 $x_2 > 95$ 是否成立? 如果成立,则打印;

(4) 如果 $x_1 > 95$ 和 $x_2 > 95$ 不能同时满足,则条件 2 不满足,此时再判条件 3;

(5) 如果 $x_1 > 95$ 不满足, 那么该学生只有在 $x_2 = 100$ 时才有可能被选中。所以, 在 $x_1 > 95$ 不满足时, 应先判 x_2 是否等于 100? 如果 $x_2 = 100$, 则再判 x_1, x_3 是否大于 80? 只要 x_1 或 x_3 中有一个不大于 80, 则该生被淘汰;

(6) 同理, 如果 $x_2 > 95$ 不满足, 那么只有在 $x_1 = 100$ 时, 该生才有可能被继续挑选。如果 $x_1 = 100$, 则再判 x_2, x_3 是否大于 80, 只要 x_2 或 x_3 中有一个不大于 80, 则该生被淘汰;

(7) 如果该生被选中, 则打印后再输入下一个学生的学号和成绩, 继续进行挑选。

根据上述设计思路, 可画出图 1-1 的流程图。

【流程图】




图 1-1